
INTRODUCTION

It is often necessary to output a floating point number
to a display. For example, to check calculations, one
might want to output floating point numbers using the
PICmicro  microcontrollers serial port, or use general
purpose I/O to output to a liquid crystal display (LCD).
Either way, the floating point number must be con-
verted to its ASCII equivalent. This document shows a
specific example of converting a 32-bit floating point
number to ASCII. Application note AN575 contains
24-bit and 32-bit floating point routines. A subroutine is
provided here that does the conversion and returns the
ASCII equivalent in RAM. An example “main” program
is provided to show how to use the subroutine.

FLOATING POINT FORMAT 
OVERVIEW

Application note AN575 describes the Microchip format
of 24 and 32-bit floating point numbers. We will use the
32-bit format for this example. Table 1 reviews the
32-bit floating point format.

TABLE 1: ARGUMENT FORMAT FOR 
MICROCHIP 32-BIT FLOATING 
POINT FORMAT 

TABLE 2: 32-BIT FLOATING POINT 
FORMAT 

Name Description

AEXP 8-bit biased exponent from -126 to +128

AARGB0 MSB of mantissa (bit7 is the sign bit)

AARGB1 middle byte of mantissa

AARGB2 LSB of mantissa

AARG

Register 
Name

AEXP
AARG

B0
AARG

B1
AARG

B2

Microchip 
32-bit format

xxxx 
xxxx

Sxxx 
xxxx

xxxx 
xxxx

xxxx 
xxxx

Authors: Rick Evans
Microchip Technology, Inc.
  1997 Microchip Technology Inc.
Table 2 depicts Microchip’s 32-bit floating point register
RAM usage. The bit labeled “S” is the sign bit. These
registers are collectively called AARG. The floating
point routines require that the arguments be put in
AARG and BARG (BARG is the second argument,
same format as AARG). The result of the floating point
operation is stored in AARG.

Floating Point to ASCII base 10 Conversion

Floating point numbers generated by the AN575 sub-
routines sometimes need to be displayed. According to
AN575, the number range for the floating point num-
bers is: ±1.17549435x10-38 to ±6.80564693x10+38.
This application note will only show how to convert
numbers between 0.000 to 9.999. With modification,
this method can be extended to convert other ranges of
numbers as well.

The calling program should ensure that the AARG reg-
isters are loaded with the correct 32-bit floating point
number: either as a result of a previous floating point
operation or by manually loading the AARG. The “main”
routine that calls float_ascii is shown in
Appendix A. For demonstration purposes, lets take an
approximation of π and load it into the AARG register.
We’ll use the number 3.1415927. (A shortcut to deter-
mine the Microchip floating point numbers is to use
fprep.exe. The program fprep.exe is provided with
AN575 to convert a decimal number to Microchip float-
ing point.) Then, the float_ascii subroutine is called.
Upon return from the subroutine, the ASCII base 10
representation of the floating point number is stored in
RAM registers: ones, tenths, hundredths, and
thousandths. Each register (ones, tenths, etc.) has an
ASCII character which represents a digit. The decimal
point is not included in the register RAM. Since it is
given that the number is between 0.000 and 9.999, the
display routine should manually output a decimal point
after it outputs the first digit. Table 3 shows the ASCII
values of each digit. The numbers are 3.141.

TABLE 3: THE ASCII VALUES FOR 3.141 
DECIMAL RETURNED FROM 
ROUTINE float_ascii 

Register Name ASCII

ones 33h

tenths 31h

hundredths 34h

thousandths 31h

Customizing the Routine

There are several changes you can make to the
float_ascii routine to customize it. First, the number
of significant figures in the number is specified by the
constant SIG_FIG. Suppose we wanted to display one
more digit of accuracy, four digits to the right of the dec-
imal point. It is easy to alter the floasc.inc assembly
file to account for this change. The following steps illus-
trate how to change the source code to return a total of
five digits.

1. Ensure that there is enough RAM registers allo-
cated to hold each digit. In this case, we would
change the cblock definition as in Figure 1.

FIGURE 1: CHANGING CBLOCK TO HOLD 
FIVE DIGITS

2. The last_digit constant must be changed.
This constant contains the address of the last
variable in the cblock. In this case, the variable
digit5 is the last location.

        last_digit set digit5

3. Now the constant, SIG_FIG should be equated
to the number of digits desired. For example, if
we desire four digits to the right of the decimal
point, there are a total of five digits that must be
obtained.

         SIG_FIG equ   5

4. Load BARG with ten thousand. Use fprep.exe
to find the floating point hexadecimal equivalent
of 10,000.

FIGURE 2: LOADING BARG WITH 10,000

cblock           ;reserve five bytes of 
    ones         ;data RAM for each digit
    tenths       ;
    hundredths   ;
    thousandths  ;
    digit5       ;**** Add one more 
                 ;RAM register
endc

movlw  0x8C    ;BARG = 10,000 decimal
movwf  BEXP    ;(floating point) fprep.exe
movlw  0x14    ;was used to get this
movwf  BARGB0  ;floating point
movlw  0x40    ;representation
movwf  BARGB1
movlw  0x00
movwf  BARGB2
AN00670A-page 2
SUMMARY

This document demonstrated converting a limited
range of the floating point numbers to ASCII. This is
useful in order to display the results of some floating
point operation. An example application of this code
could be with the PIC14C000 microcontroller. Using
the analog-to-digital converter module, one could read
the voltage on a pin from 0.000 to 3.500 volts and out-
put the decimal number to an LCD.
  1997 Microchip Technology Inc.

APPENDIX A:
list p=16c74a,st=off,mm=off

#define P16_MAP1 0
#define P16_MAP2 1

        include “p16c74a.inc”   
        nolist    
        include “math16.inc”                    ;constants and varible definitions
        list
        cblock  0x70                            ;set cblock start address for
        endc                                    ;float_ascii routine

        org     0x000
        goto    start

        org     0x005
start

        movlw   0x80                            ;     1 2345678
        movwf   AEXP                            ;PI = 3.1415927
        movlw   0x49
        movwf   AARGB0
        movlw   0x0F
        movwf   AARGB1
        movlw   0xDB
        movwf   AARGB2  
        
        call    float_ascii                     ;convert a 32-bit float to ASCII
                                                ;in this case
                                                ;ones = 3
                                                ;tenths = 1
                                                ;hundredths = 4
                                                ;thousandths = 1

done    goto    done

        include “floasc.inc”
        

        end
  1997 Microchip Technology Inc. AN00670A-page 3

;****************************************************************************
;*Floating Point to ASCII
;*
;* INPUT: 32 bit floating point number in AEXP, AARGB0, AARGB1, AARGB2
;*        For this example, the number must be between 0.000 and 9.999.
;*        (You can easily change the number range to suit your needs.)
;*
;* OUTPUT: ones, tenths, hundredths, thousandths (ASCII digit in each)
;*         
;*
;* USED: INDF,FSR,AARG,BARG,REMB,digit_count
;*
;* PROCEDURE: The floating point number in AARG is multiplied by 1000.
;*            Then the product is divided by 10 three times.  After each
;*            divide, the remainder is kept.  
;*
;*            After each digit is obtained, 30H is added to it to make it
;*            an ASCII representation of that number.  Then, the ASCII
;*            value is stored in register RAM at the locations specified
;*            in the “cblock.”      
;*
;*Note: The ASCII decimal point is not generated by this routine.
;*            You must output the decimal point in the correct decimal
;*            position for your application.  For this example, the
;*            decimal point is after the first digit: ones.
;*            
;* The following files are needed for this routine to function.
;*  p16c74a.inc-- or any other midrange processor include file 
;*                         include the processor file in your “main” file
;*
;*  math16.inc          -- constant and variable definitions for 
;*                         AN575 floating point routines and
;*                         AN617 fixed point routines, both are used
;*                         in this float to ASCII routine
;*                         “include” this file in your “main” program
;*
;*  fxd26.a16 -- fixed point 32/16 divide, included at the end
;*               of this routine.
;*
;*  fp32.a16-- 32 bit float to 32 bit integer conversion 
;*             included at the end of this program. 
;* 
;****************************************************************************

;RAM Register Definitions       Your “main” program must have a “cblock”
                                ;directive with a RAM address so the 
                                ;following “cblock” will be located in RAM

      cblock                    ;reserve four bytes of data RAM for 
        ones                    ;each digit        
        tenths        
        hundredths        
        thousandths              
      
      endc

last_digit set thousandths

      cblock
        digit_count             ;counter used to cycle through each digit 
      endc

SIG_FIG equ     4               ;set SIG_FIG equal to the number of 
AN00670A-page 4   1997 Microchip Technology Inc.

                                ;significant figures in your decimal number
                                ;for example: ones, tenths,hundredths, 
                                ;thousandths, requires 4 sig figs

float_ascii

movlw 0x88 ;BARG= 1000 decimal (floating point)
movwf BEXP            ;fprep.exe was used to get this
movlw 0x7A            ;floating point representation of 1000
movwf BARGB0
movlw 0x00
movwf BARGB1
movlw 0x00
movwf BARGB2

call FPM32 ;AARG = AARG * 1000
   

call INT3232 ;AARG  <--  INT( AARG )

movlw last_digit
movwf FSR ;pointer = address of smallest digit
movlw SIG_FIG         ;load counter with the number of
movwf digit_count     ;significant figures the decimal number

flo_asclp
clrf BARGB0 ;Make the divisor 10.  
movlw d’10’
movwf BARGB1

call FXD3216U ;divide (32-bit fixed) / 10 (to get remainder)

movf REMB1,w ;put remainder in w register
movwf INDF ;put number into appropriate digit position

movlw 0x30
addwf INDF,f ;add 30h to decimal number to convert to ASCII

decf FSR,f ;move pointer to next digit 

decfsz digit_count,f
goto flo_asclp

return

        nolist

        include “fxd26.a16”     ;fixed point 32/16 divide from AN617

        include “fp32.a16”      ;32 bit float routines
                                ;we are using FPM32 for 32-bit multiply
                                ;and INT3232 for 32-bit float to 32-bit int
                                ;conversion.  Routines are in AN575 
    
  1997 Microchip Technology Inc. AN00670A-page 5

